

**ГОСУДАРСТВЕННЫЙ КОМИТЕТ СВЯЗИ, ИНФОРМАТИЗАЦИИ И  
ТЕЛЕКОММУНИКАЦИОННЫХ ТЕХНОЛОГИЙ РЕСПУБЛИКИ  
УЗБЕКИСТАН  
ТАШКЕНТСКИЙ УНИВЕРСИТЕТ ИНФОРМАЦИОННЫХ  
ТЕХНОЛОГИЙ**

**ФАКУЛЬТЕТ «ТЕЛЕКОММУНИКАЦИОННЫЕ ТЕХНОЛОГИИ»**

**МОДЕЛИРОВАНИЕ И СИМУЛЯЦИЯ СЕТЕЙ ПЕРЕДАЧИ ДАННЫХ**

**Методические указания**

**Ташкент 2014**

Автор У.Б.Амирсаидов. «Моделирование и симуляция сетей передачи данных».- Ташкент: ТУИТ. 2014, 58 с.

Настоящие методические указания предназначены для магистрантов по специальности 5А350101 - «Телекоммуникационная инженерия».

Предлагаемый практикум может быть использован для обучения магистрантов методам построения дискретных имитационных моделей сетей, систем и устройств передачи информации средствами специального языка имитационного моделирования GPSS (General Purpose Simulation System) World.

Рассмотрены и рекомендованы к изданию на заседании научно – методического совета ТУИТ (протокол № \_\_ от \_\_\_\_\_ 2014 г.).

Рецензент:

ТУИТ, кафедра ТСП, профессор

Р.И. Исаев

Ташкентский университет информационных технологий, 2014

## ВВЕДЕНИЕ

Процессы функционирования различных систем и устройств передачи информации, вычислительных систем и сетей связи могут быть представлены той или иной совокупностью систем массового обслуживания (СМО) - стохастических, динамических, дискретно-непрерывных математических моделей.

В самом деле, на вход системы передачи информации поступают входные потоки задач, которые требуют для своего решения определенных ресурсов. Моменты поступления задач в общем случае не детерминированы. Случайными также являются и требуемые задачами ресурсы: объем памяти, время процессоров, устройства ввода/вывода и т.д. В системе могут возникать отказы и сбои затребованных ресурсов, что также является случайными событиями.

Проектировщик, меняя конфигурацию системы: количество и типы процессоров, банков памяти, интерфейсов, организацию системы, принципы и стратегии обработки и передачи информации может получать с помощью разработанной модели массового обслуживания временные характеристики проектируемой системы и характеристики для оценки ее качества обслуживания. Полученные характеристики являются для проектировщика основой для выбора наиболее эффективной конфигурации системы, принципов и стратегий обработки и передачи информации.

Исследование характеристик таких моделей может проводиться либо аналитическими методами, либо путем имитационного моделирования [1-4].

Аналитические методы моделирования рассматриваются в лекционных занятиях, а в практических занятиях, на основе данных методических указаний выполняются работы по имитационному моделированию систем массового обслуживания различного типа с использованием пакета GPSS World.

## **ПРАКТИЧЕСКАЯ РАБОТА № 1.**

### **СИСТЕМА ИМИТАЦИОННОГО МОДЕЛИРОВАНИЯ GPSS**

**Цель работы.** Изучение операторов системы имитационного моделирования GPSS. Составление простых программ на GPSS и анализ выходных статистических данных.

#### **Основные теоретические сведения**

Исходная программа на языке GPSS (GPSS/PC, GPSS WORLD), как и программа на любом языке программирования, представляет собой последовательность операторов[5,6]. Операторы GPSS записываются и вводятся в ПК в следующем формате:

номер строки, имя операции, операнды и комментарии.

Все операторы исходной программы должны начинаться с номера строки - целого положительного числа от 1 до 9999999. После ввода операторов они располагаются в исходной программе в соответствии с нумерацией строк.

В GPSS WORLD номер строки можно не писать для лучшей читаемости текста.

В поле операции записывается ключевое слово (название оператора), указывающее конкретную функцию, выполняемую данным оператором.

Каждый оператор GPSS относится к одному из четырех типов: операторы-блоки, операторы определения объектов, управляющие операторы и операторы-команды.

## Блоки, связанные с транзактами

С транзактами связаны блоки создания, уничтожения, задержки транзактов, изменения их атрибутов и создания копий транзактов.

Для создания транзактов, входящих в модель, служит блок GENERATE (генерировать), имеющий следующий формат:

имя GENERATE A,B,C,D,E.

В поле A задается среднее значение интервала времени между моментами поступления в модель двух последовательных транзактов. Если этот интервал постоянен, то поле B не используется. Если же интервал поступления является случайной величиной, то в поле B указывается модификатор среднего значения, который может быть задан в виде модификатора-интервала или модификатора-функции.

Например, блок

```
GENERATE 100,40
```

создает транзакты через случайные интервалы времени, равномерно распределенные на отрезке [60;140].

В поле C задается момент поступления в модель первого транзакта. Если это поле пусто или равно 0, то момент появления первого транзакта определяется операндами A и B.

Поле D задает общее число транзактов, которое должно быть создано блоком GENERATE. Если это поле пусто, то блок генерирует неограниченное число транзактов до завершения моделирования.

В поле E задается приоритет, присваиваемый генерируемым транзактам. Число уровней приоритетов неограничено, причем самый низкий приоритет - нулевой. Если поле E пусто, то генерируемые транзакты имеют нулевой приоритет.

Для присваивания параметрам начальных значений или изменения этих значений служит блок ASSIGN (присваивать), имеющий следующий формат:

имя ASSIGN A,B,C.

В поле A указывается номер или имя параметра, в который заносится значение операнда B. Если в поле A после имени (номера) параметра стоит знак + или -, то значение операнда B добавляется или вычитается из текущего содержимого параметра. В поле C может быть указано имя или номер функции-модификатора, действующей аналогично функции-модификатору в поле B блока GENERATE.

Например, блок

```
ASSIGN 5,0
```

записывает в параметр с номером 5 значение 0, а блок

```
ASSIGN COUNT+,1
```

добавляет 1 к текущему значению параметра с именем COUNT.

Для удаления транзактов из модели служит блок TERMINATE (завершить), имеющий следующий формат:

```
имя TERMINATE A.
```

Значение поля A указывает, на сколько единиц уменьшается содержимое так называемого счетчика завершений при входе транзакта в данный блок TERMINATE. Если поле A не определено, то оно считается равным 0, и транзакты, проходящие через такой блок, не уменьшают содержимого счетчика завершений.

Начальное значение счетчика завершений устанавливается управляющим оператором START (начать), предназначенным для запуска прогона модели. Поле A этого оператора содержит начальное значение счетчика завершений (см. разд. 3). Прогон модели заканчивается, когда содержимое счетчика завершений обращается в 0. Таким образом, в модели должен быть хотя бы один блок TERMINATE с непустым полем A, иначе процесс моделирования никогда не завершится.

Участок блок-схемы модели, связанный с парой блоков GENERATE-TERMINATE, называется сегментом. Простые модели могут

состоять из одного сегмента, в сложных моделях может быть несколько сегментов.

Например, простейший сегмент модели, состоящий всего из двух блоков GENERATE и TERMINATE и приведенный на рис. 1.1, в совокупности с управляющим оператором START моделирует процесс создания случайного потока транзактов, поступающих в модель со средним интервалом в 100 единиц модельного времени, и уничтожения этих транзактов. Начальное значение счетчика завершений равно 1000. Каждый транзакт, проходящий через блок TERMINATE, вычитает из счетчика единицу, и таким образом моделирование завершится, когда тысячный по счету транзакт войдет в блок TERMINATE. При этом точное значение таймера в момент завершения прогона непредсказуемо. Следовательно, в приведенном примере продолжительность прогона устанавливается не по модельному времени, а по количеству транзактов, прошедших через модель.

```
GENERATE 100,40  
TERMINATE 1  
START 1000
```

Рис. 1.1. Простой сегмент модели.

Если необходимо управлять продолжительностью прогона по модельному времени, то в модели используется специальный сегмент, называемый сегментом таймера (рис.1.2).

```
GENERATE 100,40  
TERMINATE  
GENERATE 100000  
TERMINATE 1  
START 1
```

Рис.1.2. Сегмент таймера.

Например, в модели из двух сегментов, приведенной на рис.1.2, первый (основной) сегмент выполняет те же функции, что и в предыдущем примере. Заметим, однако, что поле А блока TERMINATE в первом сегменте пусто, т.е. уничтожаемые транзакты не уменьшают содержимого счетчика завершений. Во втором сегменте блок GENERATE создаст первый транзакт в момент модельного времени, равный 100000. Но этот транзакт окажется и последним в данном сегменте, так как, войдя в блок TERMINATE, он обратит в 0 содержимое счетчика завершений, установленное оператором START равным 1. Таким образом, в этой модели гарантируется завершение прогона в определенный момент модельного времени, а точное количество транзактов, прошедших через модель, непредсказуемо.

В приведенных примерах транзакты, входящие в модель через блок GENERATE, в тот же момент модельного времени уничтожались в блоке TERMINATE. В моделях систем массового обслуживания заявки обслуживаются приборами (каналами) СМО в течение некоторого промежутка времени прежде, чем покинуть СМО. Для моделирования такого обслуживания, т.е. для задержки транзактов на определенный отрезок модельного времени, служит блок ADVANCE (задержать), имеющий следующий формат:

имя ADVANCE А,В

Операнды в полях А и В имеют тот же смысл, что и в соответствующих полях блока GENERATE. Следует отметить, что транзакты, входящие в блок ADVANCE, переводятся из списка текущих событий в список будущих событий, а по истечении вычисленного времени задержки возвращаются назад, в список текущих событий, и их продвижение по блок-схеме продолжается. Если вычисленное время задержки равно 0, то транзакт в тот же момент модельного времени переходит в следующий блок, оставаясь в списке текущих событий.

Например, в сегменте, приведенном на рис.1.3, транзакты, поступающие в модель из блока GENERATE через случайные интервалы времени, имеющие равномерное распределение на отрезке [60;140], попадают в блок ADVANCE. Здесь определяется случайное время задержки транзакта, имеющее равномерное распределение на отрезке [30;130], и транзакт переводится в список будущих событий. По истечении времени задержки транзакт возвращается в список текущих событий и входит в блок TERMINATE, где уничтожается. Заметим, что в списке будущих событий, а значит и в блоке ADVANCE может одновременно находиться произвольное количество транзактов.

```
GENERATE 100,40
ADVANCE 80,50
TERMINATE 1
```

Рис.1.3. Сегмент с оператором ADVANCE.

### **Блоки, связанные с аппаратными объектами**

Все примеры моделей, рассматривавшиеся выше, пока еще не являются моделями систем массового обслуживания, так как в них не учтена основная особенность СМО: конкуренция заявок на использование некоторых ограниченных ресурсов системы. Все транзакты, входящие в эти модели через блок GENERATE, немедленно получают возможность "обслуживания" в блоке ADVANCE, который никогда не "отказывает" транзактам во входе, сколько бы транзактов в нем не находилось.

Для моделирования ограниченных ресурсов СМО в модели должны присутствовать аппаратные объекты: одноканальные или многоканальные устройства. Одноканальные устройства создаются в текущей модели при

использовании блоков SEIZE (занять) и RELEASE (освободить), имеющих следующий формат:

имя SEIZE A  
имя RELEASE A

В поле A указывается номер или имя устройства. Если транзакт входит в блок SEIZE, то устройство, указанное в поле A, становится занятым и остаётся в этом состоянии до тех пор, пока этот же транзакт не пройдёт соответствующий блок RELEASE, освобождая устройство. Если устройство, указанное в поле A блока SEIZE, уже занято каким-либо транзактом, то никакой другой транзакт не может войти в этот блок и остаётся в предыдущем блоке. Транзакты, задержанные (заблокированные) перед блоком SEIZE, остаются в списке текущих событий и при освобождении устройства обрабатываются с учетом приоритетов и очередности поступления.

Каждое устройство имеет следующие СЧА: F - состояние устройства (0 - свободно, 1 - занято); FR - коэффициент использования в долях 1000; FC - число занятий устройства; FT - целая часть среднего времени занятия устройства.

### **Блоки для сбора статистических данных**

Модели СМО разрабатываются обычно для исследования различных характеристик, связанных с ожиданием заявок в очереди: длины очереди, времени ожидания и т.п., а в приведенных примерах очередь транзактов образуется в списке текущих событий и недоступна исследователю. Для регистрации статистической информации о процессе ожидания транзактов в модели должны присутствовать статистические объекты: очереди или таблицы.

Объекты типа очередь создаются в модели путем использования блоков - регистраторов очередей: QUEUE (стать в очередь) и DEPART (уйти из очереди), имеющих следующий формат:

имя QUEUE A,B

имя DEPART A,B.

В поле А указывается номер или имя очереди, а в поле В – число единиц, на которое текущая длина очереди увеличивается при входе транзакта в блок QUEUE или уменьшается при входе транзакта в блок DEPART. Обычно поле В пусто, и в этом случае его значение по умолчанию принимается равным 1.

Для сбора статистики о транзактах, заблокированных перед каким-либо блоком модели, блоки QUEUE и DEPART помещаются перед и после этого блока соответственно. При прохождении транзактов через блоки QUEUE и DEPART соответствующим образом изменяются следующие СЧА очередей: Q - текущая длина очереди; QM - максимальная длина очереди; QA - целая часть средней длины очереди; QC - общее число транзактов, вошедших в очередь; QZ - число транзактов, прошедших через очередь без ожидания (число "нулевых" входов); QT - целая часть среднего времени ожидания с учетом "нулевых" входов; QX - целая часть среднего времени ожидания без учета "нулевых" входов.

### **Блоки, изменяющие маршруты транзактов**

В приведенных выше примерах транзакты, выходящие из любого блока, всегда поступали в следующий блок. В более сложных моделях возникает необходимость направления транзактов к другим блокам в зависимости от некоторых условий. Эту возможность обеспечивают блоки изменения маршрутов транзактов.

Блок TRANSFER (передать) служит для передачи входящих в него транзактов в блоки, отличные от следующего. Блок имеет девять режимов

работы, из которых рассмотрим здесь лишь три наиболее часто используемых. В этих трех режимах блок имеет следующий формат:

имя TRANSFER A,B,C.

Смысл операндов в полях A, B и C зависит от режима работы блока.

В режиме безусловной передачи поля A и C пусты, а в поле B указывается имя блока, к которому безусловным образом направляется транзакт, вошедший в блок TRANSFER. Например:

TRANSFER ,FINAL.

В режиме статистической передачи операнд A определяет вероятность, с которой транзакт направляется в блок, указанный в поле C. С вероятностью -A транзакт направляется в блок, указанный в поле B (в следующий, если поле B пусто). Вероятность в поле A может быть задана непосредственно десятичной дробью, начинающейся с точки. Например, блок

TRANSFER .75,THIS,THAT

с вероятностью 0,75 направляет транзакты в блок с именем THAT, а с вероятностью 0,25 - в блок с именем THIS.

Если же поле A начинается не с десятичной точки и не содержит одного из ключевых слов - признаков других режимов работы блока, то его значение рассматривается как количество тысячных долей в вероятности передачи. Например, предыдущий блок TRANSFER можно записать также в следующем виде:

TRANSFER 750,THIS,THAT.

В режиме логической передачи в поле A записывается ключевое слово BOTH (оба). Транзакт, поступающий в блок TRANSFER, сначала пытается войти в блок, указанный в поле B (или в следующий блок, если поле B пусто), а если это не удается, т.е. блок B отказывает транзакту во входе, то в блок, указанный в поле C. Если и эта попытка неудачна, то транзакт задерживается в блоке TRANSFER до изменения условий в модели, делающего возможным вход в один из блоков B или C, причем при

одновременно возникшей возможности предпочтение отдается блоку В.  
Например:

```
TRANSFER BOTH,MET1,MET2.
```

Блок TEST (проверить) служит для задержки или изменения маршрутов транзактов в зависимости от соотношения двух СЧА. Он имеет следующий формат:

```
имя TEST X A,B,C.
```

Вспомогательный операнд X содержит условие проверки соотношения между СЧА и может принимать следующие значения: L (меньше); LE (меньше или равно); E (равно); NE (не равно); GE (больше или равно); G (больше). Поле A содержит первый, а поле B - второй из сравниваемых СЧА. Если проверяемое условие A X B выполняется, то блок TEST пропускает транзакт в следующий блок. Если же это условие не выполняется, то транзакт переходит к блоку, указанному в поле C, а если оно пусто, то задерживается перед блоком TEST.

Например, блок

```
TEST LE P$TIME,C1
```

не пропускает транзакты, у которых значение параметра с именем TIME больше текущего модельного времени. Блок

```
TEST L Q$LINE,5,OUT
```

направляет транзакты в блок с именем OUT, если текущая длина очереди LINE больше либо равна 5.

Для задержки или изменения маршрута транзактов в зависимости от состояния аппаратных объектов модели служит блок GATE (впустить), имеющий следующий формат:

```
имя GATE X A,B
```

Вспомогательный операнд X содержит код состояния проверяемого аппаратного объекта, а в поле A указывается имя или номер этого объекта. Если проверяемый объект находится в заданном состоянии, то блок GATE пропускает транзакт к следующему блоку. Если же заданное в блоке условие

не выполняется, то транзакт переходит к блоку, указанному в поле В, а если это поле пусто, то задерживается перед блоком GATE.

Операнд X может принимать следующие значения: U (устройство занято); NU (устройство свободно); I (устройство захвачено); NI (устройство не захвачено); SE (МКУ пусто); SNE (МКУ не пусто); SF (МКУ заполнено); SNF (МКУ не заполнено); LS (ЛП включен), LR (ЛП выключен).

Например, блок

GATE SNE BUF3

отказывает во входе транзактам, поступающим в моменты, когда в МКУ с именем BUF3 все каналы обслуживания свободны. Блок

GATE LR 4,BLOK2

направляет транзакты в блок с именем BLOK2, если в момент их поступления ЛП с номером 4 включен.

### **Блоки, работающие с памятью**

Для хранения в памяти отдельных числовых значений и массивов таких значений используются сохраняемые величины и матрицы сохраняемых величин.

Сохраняемые величины могут использоваться в модели для хранения исходных данных, которые надо изменять при различных прогонах модели, промежуточных значений и результатов моделирования. В начале моделирования все сохраняемые величины устанавливаются равными 0. Для установки отличных от 0 начальных значений сохраняемых величин используется оператор INITIAL, имеющий следующий формат:

INITIAL X\$ имя , значение;

INITIAL X j , значение.

Здесь имя и j - соответственно имя и номер сохраняемой величины, а значение - присваиваемое ей начальное значение (константа).

Для изменения сохраняемых величин в процессе моделирования служит блок SAVEVALUE (сохранить величину), имеющий следующий формат:

имя SAVEVALUE A,B.

В поле A указывается номер или имя сохраняемой величины, в которую записывается значение операнда B. Если в поле A после имени (номера) сохраняемой величины стоит знак + или - , то значение операнда B добавляется или вычитается из текущего содержимого сохраняемой величины. Например:

SAVEVALUE 5,Q\$LINE

SAVEVALUE NREF+,1.

### **Управляющие операторы GPSS**

Для управления прогоном модели используются управляющие операторы GPSS. С одним из них - оператором START - мы уже сталкивались при рассмотрении блока TERMINATE. Оператор START (начать) имеет следующий формат:

START A,B,C,D.

Поле A содержит константу, задающую начальное значение счетчика завершений. В поле B может быть записано ключевое слово NP - признак подавления формирования стандартного отчета по завершении моделирования. Если поле B пусто, то по окончании прогона модели формируется отчет со стандартной статистической информацией о всех объектах модели. Поле C не используется и сохранено для совместимости со старыми версиями GPSS. Поле D может содержать 1 для включения в отчет списков текущих и будущих событий. Если поле D пусто, то выдача в отчет содержимого этих списков не производится.

Оператор SIMULATE (моделировать) устанавливает предел реального времени, отводимого на прогон модели. Если прогон не завершится до

истечения этого времени, то он будет прерван принудительно с выдачей накопленной статистики в отчет.

Оператор `SIMULATE` имеет единственный операнд `A`, содержащий предельное время моделирования в минутах, задаваемое константой. Оператор размещается перед оператором `START`, начинающим лимитированный прогон.

Оператор `RESET` (сбросить) сбрасывает всю статистическую информацию, накопленную в процессе прогона модели. При этом состояние аппаратных, динамических и запоминающих объектов, а также генераторов случайных чисел сохраняется, и моделирование может быть возобновлено с повторным сбором статистики. Оператор не имеет операндов.

### **Получение и интерпретация стандартного отчета**

По завершении прогона модели раздается звуковой сигнал, и в строке состояния появляются сообщения

`Writing REPORT.GPS Simulation Complete Reporting ...`, сигнализирующие о том, что моделирование закончено и в данный момент производится создание отчета о прогоне модели. Затем система переходит в состояние ожидания дальнейших команд.

Отчет, создаваемый по завершении моделирования, записывается в файл со стандартным именем `REPORT.GPS`. Это имя может быть изменено командой `REPORT` (создать отчет), имеющей следующий формат:

`REPORT A,B.`

В поле `A` указывается спецификация файла, в который должен быть выведен отчет. Если поле `B` содержит ключевое слово `NOW`, то отчет создается немедленно после ввода команды.

Необходимо иметь в виду, что отчет, создаваемый автоматически по завершении прогона модели или командой `REPORT`, является неформатированным, т.е. непригодным для непосредственного просмотра.

Для форматирования и создания стандартного отчета GPSS необходимо завершить сеанс и выполнить программу форматирования отчета. Выход из интегрированной среды (завершение сеанса) производится путем ввода управляющего оператора END (закончить). При этом производится выход в MS DOS или в программу-оболочку Norton Commander.

Для форматирования отчета необходимо загрузить модуль форматирования GPSSREPT.EXE. После его загрузки на экране появляется "заставка" с названием модуля, двумя окнами в нижней части экрана и сообщениями-подсказками. В левом окне выведено имя файла, в котором находится неформатированный отчет (по умолчанию - это файл REPORT.GPS). В правом окне выведено обозначение устройства, куда должен быть выведен форматированный отчет (по умолчанию - это экран дисплея SCRN). Форматированный отчет может быть также выведен на печать или на диск. Для этого в правое окно надо ввести обозначение PRN или имя файла на диске соответственно. Для переключения окон используется клавиша Enter. Для создания отчета на выбранном устройстве следует нажать клавишу Пробел, для выхода из программы клавишу Esc.

Если содержимое окон по умолчанию оставлено без изменения, то после нажатия клавиши Пробел на экране появляется отчет о последнем прогоне модели, выполненном перед завершением сеанса работы с модулем GPSSPC.EXE. Отчет содержит следующую информацию:

- общие сведения о модели и ее прогоне, включающие модельное время начала (START\_TIME) и конца (END\_TIME) прогона, количество блоков в модели (BLOCKS), количество устройств (FACILITIES), количество многоканальных устройств (STORAGES), объем памяти, оставшейся свободной при прогоне модели (FREE\_MEMORY);

- сведения об именах объектов модели, включающие для каждого имени идентификатор (NAME), присвоенное ему числовое значение (VALUE) и тип имени: 0, если числовое значение имени присвоено пользователем с

помощью оператора EQU; 1, если числовое значение имени присвоено системой; 2, если имя является именем блока;

- сведения о блоках модели, включающие для каждого блока номер строки исходной программы (LINE), номер или имя блока (LOC), название блока (BLOCK\_TYPE), количество транзактов, прошедших через блок (ENTRY\_COUNT), текущее количество транзактов в блоке в момент завершения моделирования (CURRENT\_COUNT), количество транзактов, заблокированных перед блоком в момент завершения моделирования (RETRY0);

- сведения об устройствах модели, включающие для каждого устройства его имя или номер (FACILITY), количество занятий устройства (ENTRIES), коэффициент использования (UTIL.), среднее время на одно занятие (AVE.\_TIME) и ряд других данных;

- сведения о многоканальных устройствах модели, включающие для каждого МКУ его имя или номер (STORAGE), емкость (CAP), количество свободных каналов в момент завершения моделирования (REMAIN), наименьшее (MIN) и наибольшее (MAX) количество занятых каналов в процессе моделирования, количество занятий МКУ (ENTRIES), среднее количество занятых каналов (AVE.C), коэффициент использования (UTIL) и ряд других данных;

- сведения об очередях модели, включающие для каждой очереди ее имя или номер (QUEUE), максимальную длину очереди в процессе моделирования (MAX), текущую длину очереди в момент завершения моделирования (CONT), общее количество транзактов, вошедших в очередь в процессе моделирования (ENTRIES), и количество "нулевых" входов в очередь (ENTRIES(0)), среднюю длину очереди (AVE.CONT.), среднее время ожидания в очереди с учетом всех транзактов (AVE.TIME) и без учета "нулевых" входов (AVE.(-0));

- сведения о статистических таблицах модели, включающие для каждой таблицы ее имя или номер (TABLE), среднее значение (MEAN) и

среднеквадратическое отклонение ( STD.DEV) табулируемой величины, границы частотных интервалов ( RANGE ), частоты ( FREQUENCY) и накопленные частоты в процентах ( CUM.%) попадания наблюдений в эти интервалы.

Разумеется, сведения об объектах того или иного типа появляются в отчете только в том случае, если в модели присутствует хотя бы один объект данного типа.

## **ЗАДАНИЯ**

1. Запустить программу рис. 1.1 и получить выходные статистические данные.
2. Изменить исходные данные программы рис.1.2 и получить выходные статистические данные.
3. Изменить программу 1.3 для моделирования системы с другими параметрами равномерного распределения.

## **СОДЕРЖАНИЕ ОТЧЕТА**

1. Разработанные программы моделирования систем.
2. Листинги выходных статистических данных.
3. Результаты анализа выходных статистических данных.

## **КОНТРОЛЬНЫЕ ВОПРОСЫ**

1. Назначение SPSS.
2. Принципы создания моделей SPSS.
3. Классификация операторов SPSS.
4. Принципы создания, задержки и уничтожения транзактов.
5. Методы изменения маршрута транзакта.

6. Назначение управляющих операторов.
7. Принципы сбора и анализа статистических данных.

## **ПРАКТИЧЕСКАЯ РАБОТА №2.**

### **МОДЕЛИРОВАНИЕ ОДНОКАНАЛЬНЫХ СМО**

**Цель работы.** Моделирование и анализ вероятностно-временных характеристик одноканальных систем массового обслуживания.

#### **Основные теоретические сведения**

В моделях массового обслуживания рассматриваются системы, на вход которых в произвольные моменты времени поступают заявки, требующие обслуживания путем предоставления определенных ресурсов (каналов связи, процессоров, устройств памяти) на некоторое время и покидающие систему после обслуживания.

Системы массового обслуживания делятся на одноканальные – с одним обслуживающим прибором (процессором) и многоканальные - содержащие несколько обслуживающих приборов (процессоров). Для создания СМО необходимо задать:

- распределение длительности интервалов времени между заявками входного потока;
- число обслуживающих приборов;
- распределение длительности обслуживания заявок обслуживающими приборами;
- стратегию выбора заявок из очереди.

Для заданного множества параметров СМО можно получить такие характеристики системы, как: время обслуживания заявки в системе; время ожидания заявкой обслуживания (время нахождения в очереди); длина

очереди; загрузка обслуживающих приборов; длительность простоя обслуживающих приборов и т.д.

Очевидно, что поскольку как минимум один из параметров модели имеет вероятностный характер, то и характеристики СМО также будут вероятностными. То есть могут быть представлены как минимум на двух уровнях: статистического ряда и среднего значения. Обычно для характеристики типа СМО используют обозначение вида  $X/Y/k/n$ , где первый параметр обозначает закон распределения времени между поступлением заявок, второй параметр - закон распределения длительности обслуживания заявок, третий параметр - число обслуживающих приборов и четвертый параметр – объем буфера для организации очереди заявок. При задании законов распределения традиционно  $M$  обозначает экспоненциальный, а  $G$  - произвольный закон распределения. Кроме экспоненциального и произвольного законов распределения могут применяться эрланговский, нормальный и другие законы.

В теории массового обслуживания СМО могут объединяться путем передачи заявок с выходов одних на входы других СМО, образуя таким образом сети массового обслуживания. Для задания сети массового обслуживания требуется определить все входящие в нее СМО, а также определить конфигурацию связей систем и вероятности передачи заявок между ними. Естественно, что сумма вероятностей передачи заявок из любой СМО должна равняться 1.

Как видно из описания, модели массового обслуживания хорошо подходят для имитации функционирования систем передачи информации, локальных и глобальных сетей связи.

В практической работе в соответствии с вариантом магистранта исследуются одноканальные системы различных типов:  $M/M/1$ ,  $M/M/1/N$ ,  $D/D/1$ ,  $G/G/1$  и др.

Физические модели одноканальных СМО с и без ограниченным буфером представлены на рис.2.1 и 2.2.




Рис.2.1. Физическая модель одноканальной СМО с неограниченным буфером.




Рис.2.2. Физическая модель одноканальной СМО с ограниченным буфером.

В практической работе имитируется процесс работы для каждой из этих систем.

В процессе выполнения работы можно изменять интенсивность поступления заявок на вход системы, емкость буфера, а также время обслуживания заявок.

Кроме того, можно изменять закон распределения заявок на входе системы, а также их обслуживание.

В качестве выходных характеристик можно определить среднюю и максимальную длины очереди, среднее время пребывания заявки в очереди, коэффициент использования прибора, а также вероятность отказа заявки в обслуживании.

Кроме того, с помощью входных и выходных счетчиков можно подсчитать число поступивших и обслуженных заявок за время моделирования.

Для получения случайных величин с различными законами распределения в GPSS используются вычислительные объекты: переменные и функции.

Как известно, произвольная случайная величина связана со случайной величиной  $R$ , имеющей равномерное распределение на отрезке  $[0;1]$ , через свою обратную функцию распределения. Для некоторых случайных величин уравнение связи имеет явное решение, и значение случайной величины с заданным распределением вероятностей может быть вычислено через  $R$  по формуле. Так, например, значение случайной величины  $E$  с показательным (экспоненциальным) распределением с параметром  $d$  вычисляется по формуле:

$$E = -(1/d) * \ln(R).$$

Напомним, что параметр  $d$  имеет смысл величины, обратной математическому ожиданию  $E$ , а, следовательно,  $1/d$  - математическое ожидание (среднее значение) случайной величины  $E$ .

Для получения случайной величины  $R$  с равномерным распределением на отрезке  $[0;1]$  в GPSS имеются встроенные генераторы случайных чисел. Для получения случайного числа путем обращения к такому генератору достаточно записать системный СЧА RN с номером генератора, например, RN1. Правда, встроенные генераторы случайных чисел GPSS дают числа не на отрезке  $[0;1]$ , а целые случайные числа равномерно распределенные от 0 до 999, но их нетрудно привести к указанному отрезку делением на 1000.

Проще всего описанные вычисления в GPSS выполняются с использованием арифметических переменных. Они могут быть целыми и действительными. Целые переменные определяются перед началом

моделирования с помощью оператора определения VARIABLE (переменная), имеющего следующий формат:

имя VARIABLE выражение.

Здесь имя - имя переменной, используемое для ссылок на нее, а выражение - арифметическое выражение, определяющее переменную. Арифметическое выражение представляет собой комбинацию операндов, в качестве которых могут выступать константы, СЧА и функции, знаков арифметических операций и круглых скобок. Следует заметить, что знаком операции умножения в GPSS/PC является символ # (номер). Результат каждой промежуточной операции в целых переменных преобразуется к целому типу путем отбрасывания дробной части, и, таким образом, результатом операции деления является целая часть частного.

Действительные переменные определяются перед началом моделирования с помощью оператора определения FVARIABLE, имеющего тот же формат, что и оператор VARIABLE. Отличие действительных переменных от целых заключается в том, что в действительных переменных все промежуточные операции выполняются с сохранением дробной части чисел, и лишь окончательный результат приводится к целому типу отбрасыванием дробной части.

Арифметические переменные обоих типов имеют единственный СЧА с названием V, значением которого является результат вычисления арифметического выражения, определяющего переменную. Вычисление выражения производится при входе транзакта в блок, содержащий ссылку на СЧА V с именем переменной.

Действительные переменные могут быть использованы для получения случайных интервалов времени с показательным законом распределения. Пусть распределения времени поступления транзактов и времени задержки должны иметь показательный закон. Это может быть сделано так, как показано на рис.2.3.

```

TARR FVARIABLE -100#LOG((1+RN1)/1000)
TSRV FVARIABLE -80#LOG((1+RN1)/1000)
GENERATE V$TARR
ADVANCE V$TSRV
TERMINATE 1

```

Рис.2.3. Сегмент модели с показательным законом распределения.

Переменная с именем TARR задает выражение для вычисления интервала поступления со средним значением 100, вторая переменная с именем TSRV - для вычисления времени задержки со средним значением 80. Блоки GENERATE и ADVANCE содержат в поле A ссылки на соответствующие переменные, при этом поле B не используется, так как в поле A содержится случайная величина, не нуждающаяся в модификации.

Большинство случайных величин не может быть получено через случайную величину R с помощью арифметического выражения. Кроме того, такой способ является достаточно трудоемким, так как требует обращения к математическим функциям, вычисление которых требует десятков машинных операций. Другим возможным способом является использование вычислительных объектов GPSS типа функция.

Функции используются для вычисления величин, заданных табличными зависимостями. Каждая функция определяется перед началом моделирования с помощью оператора определения FUNCTION (функция), имеющего следующий формат:

```
имя FUNCTION A,B.
```

Здесь имя - имя функции, используемое для ссылок на нее; A - стандартный числовой атрибут, являющийся аргументом функции; B - тип функции и число точек таблицы, определяющей функцию.

Как уже отмечалось, при использовании функции в поле B блоков GENERATE и ADVANCE вычисление интервала поступления или времени

задержки производится путем умножения операнда А на вычисленное значение функции. Отсюда следует, что функция, используемая для генерирования случайных чисел с показательным распределением, должна описывать зависимость  $y=-\ln(x)$ , представленную в табличном виде. Оператор FUNCTION с такой таблицей, содержащей 24 точки для обеспечения достаточной точности аппроксимации, имеет следующий вид:

```

EXPON FUNCTION RN1,C24
0,0/.1,.104/.2,.222/.3,.355/.4,.509/.5,.69/.6,.915
.7,1.2/.75,1.38/.8,1.6/.84,1.83/.88,2.12/.9,2.3
.92,2.52/.94,2.81/.95,2.99/.96,3.2/.97,3.5/.98,3.9
.99,4.6/.995,5.3/.998,6.2/.999,7/.9997,8

```

Вычисление непрерывной функции производится следующим образом. Сначала определяется интервал  $(X_i; X_{i+1})$ , на котором находится текущее значение СЧА-аргумента (в нашем примере - сгенерированное значение RN1). Затем на этом интервале выполняется линейная интерполяция с использованием соответствующих значений  $Y_i$  и  $Y_{i+1}$ . Результат интерполяции усечается (отбрасыванием дробной части) и используется в качестве значения функции. Если функция служит операндом В блоков GENERATE или ADVANCE, то усечение результата производится только после его умножения на значение операнда А.

Воспользуемся блоками SEIZE и RELEASE для моделирования одноканальной СМО с ожиданием (рис.2.4). Теперь блок ADVANCE находится между блоками SEIZE и RELEASE, моделирующими занятие и освобождение устройства с именем SYSTEM, и поэтому в нем может находиться только один транзакт. Транзакты, выходящие из блока GENERATE в моменты занятости устройства, не смогут войти в блок SEIZE и будут оставаться в блоке GENERATE, образуя очередь в списке текущих событий.

```

EXPON FUNCTION RN1,C24
0,0/.1,.104/.2,.222/.3,.355/.4,.509/.5,.69/.6,.915
.7,1.2/.75,1.38/.8,1.6/.84,1.83/.88,2.12/.9,2.3
.92,2.52/.94,2.81/.95,2.99/.96,3.2/.97,3.5/.98,3.9
.99,4.6/.995,5.3/.998,6.2/.999,7/.9997,8

GENERATE 100,FN$EXPON
SEIZE SYSTEM
ADVANCE 80,FN$EXPON
RELEASE SYSTEM
TERMINATE 1
START 1000

```

Рис.2.4. Модель СМО с ожиданием.

Очень часто исследователя интересует не только среднее значение времени ожидания в очереди, но и дисперсия этого времени, а также статистическое распределение выборки времени ожидания, представляемое обычно графически в виде гистограммы. Имея такое распределение, можно оценить вероятность того, что время ожидания превысит или не превысит некоторое заданное значение. Для сбора и обработки данных о выборочном распределении времени ожидания в очереди служат статистические объекты типа Q-таблица.

Для создания в модели такой таблицы она должна быть предварительно определена с помощью оператора определения QTABLE (Q-таблица), имеющего следующий формат:

```
имя QTABLE A,B,C,D.
```

Здесь имя - имя таблицы, используемое для ссылок на нее; A - номер или имя очереди, распределение времени ожидания в которой необходимо

получить;  $B$  - верхняя граница первого частотного интервала таблицы;  $C$  - ширина частотных интервалов;  $D$  - количество частотных интервалов.

Диапазон всевозможных значений времени ожидания в очереди, указанной в поле  $A$ , разбивается на ряд частотных интервалов, количество которых указано в поле  $D$ . Первый из этих интервалов имеет ширину от минус бесконечности до величины, указанной в поле  $B$ , включительно. Второй интервал включает значения, большие, чем величина первой границы в поле  $B$ , но меньшие или равные  $B+C$ , и т.д. Все промежуточные интервалы имеют одинаковую ширину, указанную в поле  $C$ . Наконец, последний интервал включает все значения, большие, чем последняя граница. Значения операндов  $B$ ,  $C$  и  $D$  должны задаваться целыми константами. Операнд  $B$  может быть неположительным, хотя для  $Q$ -таблицы это не имеет смысла, так как время не может быть отрицательным. Операнды  $C$  и  $D$  должны быть строго положительными.

При прохождении транзакта через блоки `QUEUE` и `DEPART` его время ожидания фиксируется, и к счетчику частотного интервала таблицы, в который попало это время, добавляется 1. Одновременно в таблице накапливается информация для вычисления среднего значения и среднеквадратического отклонения (корня из дисперсии) времени ожидания. По окончании моделирования среднее значение и среднеквадратическое отклонение времени ожидания, а также счетчики попаданий в различные частотные интервалы выводятся в стандартный отчет GPSS/PC. Таблицы, как и другие объекты GPSS, имеют СЧА:  $TC$  – общее число транзактов, вошедших в очередь, связанную с таблицей;  $TB$  - целая часть среднего времени ожидания в очереди;  $TD$  - целая часть среднеквадратического отклонения времени ожидания в очереди.

Дополним модель из примера на рис. 2.4 оператором `QTABLE` для получения распределения времени ожидания в очереди с именем `LINE` (рис. 2.5).

```

WTIME QTABLE  LINE,50,50,10
EXPON FUNCTION RN1,C24
0,0/.1,.104/.2,.222/.3,.355/.4,.509/.5,.69/.6,.915
.7,1.2/.75,1.38/.8,1.6/.84,1.83/.88,2.12/.9,2.3
.92,2.52/.94,2.81/.95,2.99/.96,3.2/.97,3.5/.98,3.9
.99,4.6/.995,5.3/.998,6.2/.999,7/.9997,8
GENERATE 100, FN$EXPON
QUEUE LINE
SEIZE SYSTEM
DEPART LINE
ADVANCE 80, FN$EXPON
RELEASE SYSTEM
TERMINATE 1
START 1000

```

Рис. 2.5. Модель системы с оператором QTABLE.

Оператор определения таблицы с именем WTIME разбивает ось времени на 10 частотных интервалов. Первый интервал включает значения от 0 до 50, второй - от 50 до 100, третий - от 100 до 150 и т.д. Последний, десятый, интервал включает значения, превышающие 450. Если, например, время ожидания некоторого транзакта в очереди составило 145 единиц модельного времени, то к счетчику третьего частотного интервала будет добавлена 1. Следует заметить, что информация в таблицу с именем WTIME заносится автоматически, при входе транзактов в блоки QUEUE и DEPART, и никаких специальных мер для этого принимать не требуется.

## ЗАДАНИЕ

1. Запустить программу имитационного моделирования одноканального СМО. Получить набор выходных статистических данных при различных законах распределения поступающих сообщений и свести их в таблицу 2.1.

Таблица 2.1

Характеристики одноканальных СМО

Показатель	Законы распределения поступающих сообщений			
	Детерминированный (по варианту)	Равномерный (по варианту)		Пуассона (по варианту)
		отклонение 10%	отклонение 50%	
Средняя длина очереди				
Максимальная длина очереди				
Коэффициент использования прибора				
Количество обслуженных заявок				

2. С помощью статистических данных, полученных в результате моделирования одноканальной системы с неограниченным буфером построить зависимости:

$$q = f(\lambda_{\text{вх}}), t_{\text{сист}} = f(\lambda_{\text{вх}}), Y_{\text{обсл}} = f(\lambda_{\text{вх}}).$$

При этом время обслуживания заявок –50, интервал между заявками изменять в диапазоне 25–100.

3. Построить те же зависимости, изменяя время обслуживания в интервале 25-100, для интенсивности поступления заявок, соответствующей интервалу между заявками в 50.

4. Разработать программу для одноканальной системы с ограниченным буфером и определить вероятностно-временные характеристики.

5. Экспериментальные зависимости подтвердить расчетами по формулам Эрланга.

## **СОДЕРЖАНИЕ ОТЧЕТА**

Отчет должен содержать:

1. Программы имитационного моделирования.
2. Графики зависимости параметров СМО от интенсивности поступления пакетов при различных законах распределения.
3. Сравнительный анализ экспериментальных и теоретических результатов.

## **КОНТРОЛЬНЫЕ ВОПРОСЫ**

1. Основные типы и параметры СМО.
2. Законы распределения интервалов поступления и обслуживания заявок.
3. Методы генерации случайных чисел.
4. Основные характеристики СМО с ограниченным и неограниченным буфером.

5. Операторы ограничения объема буфера.
6. Принцип определения потери транзактов из-за переполнения буфера.
7. Сравнительный анализ характеристик СМО с ограниченным и неограниченным буфером.

### ПРАКТИЧЕСКАЯ РАБОТА № 3. МОДЕЛИРОВАНИЕ МНОГОКАНАЛЬНЫХ СМО

**Цель работы.** Моделирование и анализ вероятностно-временных характеристик многоканальных систем массового обслуживания.

#### Основные теоретические сведения

Для создания в модели многоканальных устройств (процессоров) (МКУ) они должны быть предварительно определены с помощью операторов разделения STORAGE (память), имеющих следующий формат:

имя STORAGE A

Здесь имя - имя МКУ (FAC), используемое для ссылок на него; A – емкость (количество каналов обслуживания) МКУ, задаваемая константой (рис.3.1).




Рис.3.1. Модель многоканальной СМО с ограниченным буфером.

Для занятия и освобождения каналов обслуживания МКУ используется пара блоков ENTER (войти) и LEAVE (покинуть), имеющих следующий формат:

```
имя  ENTER  А,В  
имя  LEAVE  А,В.
```

В поле А указывается номер или имя МКУ, в поле В число каналов МКУ, занимаемых при входе в блок ENTER или освобождаемых при входе в блок LEAVE. Обычно поле В пусто, и в этом случае по умолчанию занимает или освобождается один канал.

При входе транзакта в блок ENTER текущее содержимое МКУ увеличивается на число единиц, указанное в поле В. Если свободная емкость МКУ меньше значения поля В, то транзакт не может войти в блок ENTER и остается в предыдущем блоке, образуя очередь в списке текущих событий.

При входе транзакта в блок LEAVE текущее содержимое МКУ уменьшается на число единиц, указанное в поле В. Не обязательно освобождается такое же число каналов МКУ, какое занималось при входе данного транзакта в блок ENTER, однако текущее содержимое МКУ не должно становиться отрицательным.

Многоканальные устройства имеют следующие СЧА: S - текущее содержимое МКУ; R - свободная емкость МКУ; SR - коэффициент использования в долях 1000; SA - целая часть среднего содержимого МКУ; SM - максимальное содержимое МКУ; SC - число занятий МКУ; ST - целая часть среднего времени занятия МКУ.

Воспользуемся блоками ENTER-LEAVE и оператором STORAGE для моделирования двухканальной СМО с ожиданием (рис.3.2). Если текущее содержимое МКУ с именем STO2 меньше 2, т.е. в блоке ADVANCE находится один или ни одного транзакта, то очередной транзакт, поступающий в модель через блок GENERATE, может войти в блок ENTER

и затем в блок ADVANCE. Если же текущее содержимое МКУ равно 2, то очередной транзакт остается в блоке GENERATE, образуя очередь в списке текущих событий. По истечении задержки одного из двух обслуживаемых транзактов в блоке ADVANCE и после входа его в блок LEAVE первый из заблокированных транзактов сможет войти в блок ENTER.

```
STO2 STORAGE 2
EXPON FUNCTION RN1,C24
0,0/.1,.104/.2,.222/.3,.355/.4,.509/.5,.69/.6,.915
.7,1.2/.75,1.38/.8,1.6/.84,1.83/.88,2.12/.9,2.3
.92,2.52/.94,2.81/.95,2.99/.96,3.2/.97,3.5/.98,3.9
.99,4.6/.995,5.3/.998,6.2/.999,7/.9997,8
GENERATE 100, FN$EXPON
ENTER STO2
ADVANCE 160, FN$EXPON
LEAVE STO2
TERMINATE 1
```

Рис. 3.2. Моделирование двухканальной СМО с ожиданием.

Таблицы в GPSS могут использоваться в более общем случае не только для табулирования времени ожидания в очереди, но и для получения выборочных распределений произвольных СЧА любых объектов модели. Для определения таблиц служит оператор TABLE (таблица), формат которого совпадает с форматом оператора QTABLE. Отличие состоит лишь в том, что в поле A оператора TABLE записывается стандартный числовой атрибут, выборочное распределение которого необходимо получить, а операнды B, C и D определяют разбиение на частотные интервалы диапазона всевозможных значений этого СЧА.

Занесение информации в таблицу, определяемую оператором TABLE, уже не может быть выполнено симулятором автоматически, как в случае Q-таблиц. Для этого используется специальный блок TABULATE (табулировать), имеющий следующий формат:

имя TABULATE A.

В поле A указывается номер или имя таблицы, определенной соответствующим оператором TABLE.

При входе транзакта в блок TABULATE текущее значение табулируемого аргумента таблицы, указанного в поле A оператора TABLE, заносится в нее в соответствии с заданным в операторе TABLE разбиением области значений аргумента на частотные интервалы. Одновременно корректируются текущие значения СЧА таблицы: счетчик входов в таблицу ТС, среднее время ожидания ТВ и среднеквадратическое отклонение времени ожидания TD.

Пусть, например, в модели многоканальной СМО, приведенной на рис. 3.2, надо получить распределение времени пребывания заявок в системе, включающего время ожидания в очереди и время обслуживания. Это может быть обеспечено способом, показанным на рис. 3.3.

Оператор TABLE определяет таблицу с именем TTIME, аргументом которой служит СЧА M1 - время пребывания транзакта в модели. В рассматриваемой модели значение СЧА M1 одновременно будет являться временем пребывания транзакта в СМО в том случае, если занесении информации в таблицу производить перед выходом транзакта из модели. Поэтому блок TABULATE, заносающий информацию о времени пребывания каждого транзакта в модели в таблицу TTIME, располагается перед блоком TERMINATE. Диапазон возможных значений времени пребывания транзакта в модели разбит в операторе TABLE на 12 частотных интервалов, ширина которых (кроме последнего) равна 100 единицам модельного времени.

```

TTIME TABLE    M1,100,100,12
STO2 STORAGE    2
EXPON FUNCTION  RN1,C24
0,0/.1,.104/.2,.222/.3,.355/.4,.509/.5,.69/.6,.915
.7,1.2/.75,1.38/.8,1.6/.84,1.83/.88,2.12/.9,2.3
.92,2.52/.94,2.81/.95,2.99/.96,3.2/.97,3.5/.98,3.9
.99,4.6/.995,5.3/.998,6.2/.999,7/.9997,8
GENERATE 100, FN$EXPON
ENTER STO2
ADVANCE 160, FN$EXPON
LEAVE STO2
TABULATE TTIME
TERMINATE 1

```

Рис.3.3. Моделирование многоканальной СМО с оператором TABLE.

Блоки рассматриваемой группы используются при моделировании различных СМО с потерями заявок. Воспользуемся, например, блоками TRANSFER для моделирования двухканальной СМО с отказами и повторными попытками (рис.3.4).

Транзакты, поступающие в модель, попадают в блок TRANSFER с именем ENT1, работающий в логическом режиме. Если в момент поступления транзакта в МКУ STO2 хотя бы один канал свободен, то блок TRANSFER направит транзакт в следующий блок, т.е. в блок ENTER. Если же в момент поступления оба канала МКУ заняты, и поэтому блок ENTER отказывает во входе, то транзакт будет направлен в блок TRANSFER с именем REFUS, работающий в статистическом режиме. С вероятностью 0,9 транзакты из этого блока передаются в следующий блок, задерживаются в нем на случайное время и с помощью блока TRANSFER, работающего в безусловном режиме, передаются вновь на вход модели в блок с именем

ENT1. С вероятностью 0,1 транзакты из блока с именем REFUS передаются в блок TERMINATE с именем OUT для уничтожения.

```
STO2 STORAGE 2
EXPON FUNCTION RN1,C24
0,0/.1,.104/.2,.222/.3,.355/.4,.509/.5,.69/.6,.915
.7,1.2/.75,1.38/.8,1.6/.84,1.83/.88,2.12/.9,2.3
.92,2.52/.94,2.81/.95,2.99/.96,3.2/.97,3.5/.98,3.9
.99,4.6/.995,5.3/.998,6.2/.999,7/.9997,8
GENERATE 100, FN$EXPON
ENT1 TRANSFER BOTH,,REFUS
ENTER STO2
ADVANCE 160, FN$EXPON
LEAVE STO2
TERMINATE 1
REFUS TRANSFER .1,,OUT
ADVANCE 250, FN$EXPON
TRANSFER ,ENT1
OUT TERMINATE 1
```

Рис.3.4. Моделирование двухканальной СМО с отказами и повторными попытками.

Следует заметить, что для уничтожения транзактов, получивших отказ в обслуживании, понадобился отдельный блок TERMINATE для фиксации в стандартном отчете количества потерянных транзактов с помощью счетчика блока с именем OUT (СЧА N\$OUT).

Для моделирования той же СМО может быть использован также блок TEST (рис.3.5). В этом варианте модели транзакт проходит в блок ENTER, если текущее число занятых каналов (СЧА S\$STO2) меньше 2.

```

STO2 STORAGE 2
EXPON FUNCTION RN1,C24
0,0/.1,.104/.2,.222/.3,.355/.4,.509/.5,.69/.6,.915
.7,1.2/.75,1.38/.8,1.6/.84,1.83/.88,2.12/.9,2.3
.92,2.52/.94,2.81/.95,2.99/.96,3.2/.97,3.5/.98,3.9
.99,4.6/.995,5.3/.998,6.2/.999,7/.9997,8
GENERATE 100,FN$EXPON
ENT1 TEST L S$STO2,2,REFUS
ENTER STO2
ADVANCE 160,FN$EXPON
LEAVE STO2
TERMINATE 1
REFUS TRANSFER .1,,OUT
ADVANCE 250,FN$EXPON
TRANSFER ,ENT1
OUT TERMINATE 1

```

Рис. 3.5. Моделирование многоканальной СМО с оператором TEST.

При использовании блока GATE модель принимает вид, показанный на рис. 3.6. В этом варианте транзакт проходит в блок ENTER, если условие "МКУ STO2 не заполнено" истинно.

```

STO2 STORAGE 2
EXPON FUNCTION RN1,C24
0,0/.1,.104/.2,.222/.3,.355/.4,.509/.5,.69/.6,.915
.7,1.2/.75,1.38/.8,1.6/.84,1.83/.88,2.12/.9,2.3
.92,2.52/.94,2.81/.95,2.99/.96,3.2/.97,3.5/.98,3.9
.99,4.6/.995,5.3/.998,6.2/.999,7/.9997,8
GENERATE 100,FN$EXPON

```

```

ENT1 GATE SNF STO2,REFUS
      ENTER STO2
      ADVANCE 160, FN$EXPON
      LEAVE STO2
      TERMINATE 1
REFUS TRANSFER .1,,OUT
      ADVANCE 250, FN$EXPON
      TRANSFER ,ENT1
OUT TERMINATE 1

```

Рис. 3.6. Моделирование многоканальной СМО с оператором GATE.

### ЗАДАНИЕ

1. Разработать программу имитационного моделирования многоканального СМО неограниченным буфером. Получить набор выходных статистических данных при различных законах распределения поступающих сообщений.

2. С помощью статистических данных, полученных в результате моделирования многоканальной системы с неограниченным буфером построить зависимости:

$$q = f(\lambda_{\text{вх}}), t_{\text{сист}} = f(\lambda_{\text{вх}}), Y_{\text{обсл}} = f(\lambda_{\text{вх}}).$$

3. Построить те же зависимости, изменяя количество каналов от 2 до 5.

4. Разработать программу для многоканальной системы с ограниченным буфером и определить вероятность отказа в обслуживании при количестве каналов от 2 до 5.

5. Экспериментальные зависимости подтвердить расчетами по формулам Эрланга.

## **СОДЕРЖАНИЕ ОТЧЕТА**

Отчет должен содержать:

1. Программы имитационного моделирования многоканальных СМО.
2. Графики зависимости параметров СМО от количества каналов и интенсивности поступления пакетов при различных законах распределения.
3. Сравнительный анализ экспериментальных и теоретических результатов.

## **КОНТРОЛЬНЫЕ ВОПРОСЫ**

1. Основные характеристики многоканальных СМО.
2. Операторы создания каналов.
3. Статистические выходные данные многоканальных устройств.
4. Организация условного выбора канала.
5. Примеры использования моделей многоканальных СМО для описания функционирования систем управления.

## **ПРАКТИЧЕСКАЯ РАБОТА №4.**

### **МОДЕЛИРОВАНИЕ ПРИОРИТЕТНЫХ СМО**

**Цель работы.** Моделирование и анализ характеристик приоритетных систем массового обслуживания.

### **Основные теоретические сведения**

Некоторые типы заявок, обрабатываемых системой передачи, имеют преимущества перед заявками других типов. Например, стоимость задержки при обработке определенного вида заявок может быть очень высокой,

вследствие чего целесообразно иметь время задержки по возможности минимальным. В других случаях предпочтение при обработке, оказываемое запросам с малым временем обслуживания, может сократить длину очереди и, тем самым, уменьшить объем буферного запоминающего устройства для хранения запросов. Пусть на обработку в систему передачи поступает несколько потоков заявок. Заявки  $i$ -го потока называют запросами приоритета  $i$ . Заявки  $i$ -го потока имеют более высокий приоритет по сравнению с заявками  $j$ -го потока ( $i < j$ ), если запросы  $i$ -го потока пользуются преимуществами перед запросами  $j$ -го потока. В соответствии с этими преимуществами, приоритеты могут быть разделены на относительные и абсолютные. Относительный приоритет имеет место, когда поступивший в СУ запрос обрабатывается без прерывания другим запросом. Следующим обрабатываемым запросом является запрос с самым высоким приоритетом или, если таких запросов несколько, один из них, тот, который ждет обслуживания дольше остальных. Абсолютный приоритет характеризуется тем, что запрос с более высоким приоритетом прерывает обслуживание запроса с низким приоритетом.

Рассмотрим GPSS-модель системы передачи, на вход которой поступают четыре пуассоновских потока заданий с относительными приоритетами и со следующими параметрами:  $X_1=0.005$ ;  $X_2=0.01$ ;  $X_3=0.03$ ;  $X_4=0.04$  заявок в ед.времени. Время обработки задания для этих четырех потоков подчиняется экспоненциальному распределению с параметром  $\mu_j=0.1$  заявок в единицу времени. Эта модель приведена рис.4.1.

В строке 1 устройству SYST назначен номер 1. Нормированное экспоненциальное распределение описано строками 2-7. Блоки с номерами 1, 3, 5 и 7 генерируют транзакты с приоритетами, равными соответственно 1, 2, 3, 4, образующими пуассоновские потоки с заданными параметрами.

```

EXPON FUNCTION  RN1,C24
0,0/.1,.104/.2,.222/.3,.355/.4,.509/.5,.69/.6,.915
.7,1.2/.75,1.38/.8,1.6/.84,1.83/.88,2.12/.9,2.3
.92,2.52/.94,2.81/.95,2.99/.96,3.2/.97,3.5/.98,3.9
.99,4.6/.995,5.3/.998,6.2/.999,7/.9997,8
    GENERATE  200,FN$EXPON,,,4
    TRANSFER  ,MET1
    GENERATE  100,FN$EXPON,,,3
    TRANSFER  ,MET1
    GENERATE  50,FN$EXPON,,,2
    TRANSFER  ,MET1
    GENERATE  25,FN$EXPON,,,1
MET1 QUEUE  PR
    SEIZE  SYST
    DEPART  PR
    ADVANCE  10,FN$EXPON
    RELEASE  SYST
    TERMINATE  1
    START  1000

```

Рис.4.1. СМО с относительным приоритетом.

Значения приоритетов в GPSS располагаются в порядке, обратном принятому в теории массового обслуживания ( $1 < 2 < 3 < 4$ ). Блок 8 выполняет присваивание параметру 1 транзакта значения, соответствующего номеру устройства SYST, следовательно, очереди будет присвоен номер 1 (см. блоки 9 и 11).

## ЗАДАНИЕ

1. Запустить программу имитационного моделирования СМО с относительным приоритетом. Получить набор выходных статистических данных при различных законах распределения поступающих сообщений.

2. Разработать программу имитационного моделирования СМО с абсолютным приоритетом. Получить набор выходных статистических данных при различных законах распределения поступающих сообщений.

3. С помощью статистических данных, полученных в результате моделирования СМО с относительным и абсолютными приоритетами построить зависимости:

$$q = f(\lambda_{\text{вх}}), t_{\text{сист}} = f(\lambda_{\text{вх}}), Y_{\text{обсл}} = f(\lambda_{\text{вх}}).$$

4. Произвести сравнительный анализ вероятностно-временных характеристик СМО с относительным и абсолютным приоритетами.

## СОДЕРЖАНИЕ ОТЧЕТА

Отчет должен содержать:

1. Программы имитационного моделирования СМО с относительным и абсолютным приоритетами.

2. Графики зависимости параметров СМО от количества приоритетов и интенсивности поступления пакетов при различных законах распределения.

3. Результаты сравнительного анализа вероятностно-временных характеристик СМО с относительным и абсолютным приоритетами.

## КОНТРОЛЬНЫЕ ВОПРОСЫ

1. Методы приоритетного обслуживания.

2. Характеристики СМО с относительным приоритетом.
3. Характеристики СМО с абсолютным приоритетом.
4. Организация приоритетного обслуживания в GPSS.
5. Сравнительный анализ характеристик СМО с относительным и абсолютным приоритетами.
6. Примеры использования моделей СМО с приоритетами для описания систем управления.

## **ПРАКТИЧЕСКАЯ РАБОТА № 5.**

### **МОДЕЛИРОВАНИЕ СМО С ИЗМЕНЯЕМОЙ ИНТЕНСИВНОСТЬЮ ОБСЛУЖИВАНИЯ**

**Цель работы.** Моделирование и анализ вероятностно-временных характеристик СМО с изменяемой интенсивностью обслуживания

#### **Основные теоретические сведения**

Некоторые системы управления (СУ) и передачи информации могут быть представлены как управляемые СМО, то есть СМО, в которых интенсивность обслуживания изменяется в зависимости от числа запросов, имеющих в СМО.

Рассмотрим модель СУ, решающей задачи оперативного управления некоторым объектом. С объекта управления на СУ поступают информационные заявки, интенсивность обработки которых зависит от общего числа заявок, находящихся на обслуживании в СУ. При числе заявок больше десяти СУ производит укороченную обработку каждой заявки за время равное 100 единиц, в противном случае обработка выполняется за 600 единиц реального времени. Считается, что в этом режиме СУ обеспечивает безаварийную работу управляемого объекта. Информационные заявки

поступают на вход СУ в случайные равномерно распределенные в интервале [200,600] моменты времени.

Модель СУ с изменяемой интенсивностью обслуживания представлена на рис.5.1. Для имитации переменного времени обслуживания информационных посылок в модели используется блок ADVANCE, меняющий время задержки в зависимости от значения СЧА Q\$ETM1. Зависимость времени задержки от СЧА Q\$ETM1 задана дискретной функцией PRER (строки 1-2). При числе транзактов, находящихся между блоками QUEUE ETM1 и DEPART ETM1, не превышающем 10, блок ADVANCE осуществляет задержку транзакта на 600 единиц, иначе - на 100 единиц.

```
PRER FUNCTION Q$ETM1,D2
      10,600/200,100
GENERATE 400,200
QUEUE   ETM1
EIZE    SYST
DEPART  ETM1
ADVANCE FN$PRER
RELEASE SYST
TERMINATE 1
START  1000
```

5.1. Моделирование СМО с изменяемой интенсивностью обслуживания.

Назначение других блоков и команд модели было рассмотрено выше. После прогона модели на ЭВМ на печать будут выданы статистика об устройстве SYST, имитирующем СУ, и очередь ETM1.

Рассмотрим модель СУ со случайным выбором программ обработки информационных заявок. Пусть имеется 3 типа информационных заявок, каждая из которых обрабатывается по своей программе. Известно, что время обработки заявки 1-го типа составляет 100 единиц времени, 2-го - 200, и 3-го - 1000. Выбор программы обработки определяется типом заявки. Так как заранее неизвестно, какая именно заявка поступает на обработку, то

осуществляется случайный выбор одной из программ. Задана вероятность поступления заявок различных типов: 1-го типа - 0.4, 2-го - 0.55, 3-го - 0.05. Требуется собрать статистику о системе при таком же потоке обрабатываемых посылок, как и в предыдущем примере. Модель для решения этой задачи представлена на рис.5.2.

```
DISR FUNCTION RN5,D3
0.4,100/0.95,200/1.,1000
GENERATE 400,200
QUEUE ETM1
SEIZE SYST
DEPART ETM1
ADVANCE FN$DISR
RELEASE SYST
TERMINATE 1
START 1000
```

Рис.5.2. Моделирование СМО со случайным выбором программ обработки.

Для случайного выбора программ обработки в модели используется дискретная случайная функция DISR, принимающая значения 100 с вероятностью 0.4, значения 200 с вероятностью 0.55 и 1000 с вероятностью 0.05 (строки 1-2). Функция DISR использует значения случайных величин, вырабатываемых программным датчиком GPSS RN5. ВС имитируется устройством с именем SYST. Блок ADVANCE осуществляет задержку обрабатываемой заявки (транзакта) в зависимости от ее типа в соответствии со значением функции DISR.

## **ЗАДАНИЕ**

1. Запустить программы 5.1 и 5.2. Получить набор выходных статистических данных.

2. Изменить программу 5.1 и 5.2 для моделирования рассматриваемых систем с экспоненциальным законом распределения поступающих заявок и получить набор выходных статистических данных при различных законах распределения поступающих заявок.

3. С помощью статистических данных, полученных в результате моделирования построить зависимости вероятностно-временных характеристик систем при различных законах распределения поступающих заявок.

## **СОДЕРЖАНИЕ ОТЧЕТА**

Отчет должен содержать:

1. Программы имитационного моделирования систем с изменяемой интенсивностью обслуживания задач и случайным выбором программы обработки.

2. Графики зависимости вероятностно-временных характеристик систем при различных законах распределения поступающих заявок.

3. Результаты анализа вероятностно-временных характеристик систем с изменяемой интенсивностью обслуживания задач и случайным выбором программы обработки.

## **КОНТРОЛЬНЫЕ ВОПРОСЫ**

1. Операторы для изменения интенсивности обслуживания.
2. Операторы для случайного выбора программы обработки.
3. Принципы формирования дискретных функций.

4. Примеры использования моделей управляемых СМО для описания систем управления.

## **ПРАКТИЧЕСКАЯ РАБОТА № 6.**

### **МОДЕЛИРОВАНИЕ МНОГОФАЗНЫХ СМО**

**Цель работы.** Моделирование и анализ вероятностно-временных характеристик последовательных и параллельных систем массового обслуживания.

#### **Основные теоретические сведения**

Рассмотрим модель системы, состоящей из 1-го процессора предварительной обработки информационных заявок и 2-го процессора, входным потоком заданий для которого служит выходной поток первого. Такая система осуществляет последовательную обработку информации и относится к многофазовым СМО. На рис.6.1 представлена GPSS-модель этой системы.

```
COMP1 EQU    1
COMP2 EQU    2
TABT TABLE  M1,0,100,1000
EXPON FUNCTION RN1,C24
0,0/.1,.104/.2,.222/.3,.355/.4,.509/.5,.69/.6,.915
.7,1.2/.75,1.38/.8,1.6/.84,1.83/.88,2.12/.9,2.3
.92,2.52/.94,2.81/.95,2.99/.96,3.2/.97,3.5/.98,3.9
.99,4.6/.995,5.3/.998,6.2/.999,7/.9997,8
GENERATE 100,FN$EXPON
ASSIGN 1,COMP1
QUEUE P1
```

```

SEIZE  COMP1
DEPART P1
ADVANCE 50, FN$EXPON
RELEASE COMP1
ASSIGN 2, COMP2
QUEUE  P2
SEIZE  COMP2
DEPART P2
ADVANCE 50,40
RELEASE COMP2
TABULATE TABT
TERMINATE 1
START 1000

```

Рис.6.1. Моделирование многофазной СМО.

Входной поток заданий является пуассоновским с параметром  $\lambda=0,01$  заявок в ед.времени. Время обработки задания для 1-го процессора распределено по экспоненциальному закону, для 2-го - по равномерному в интервале  $[10,90]$ . Результатом прогона модели являются статистика относящиеся к обоим процессорам и очередям к ним и таблица частот для времени обработки задания..

Моделирование параллельных систем рассмотрим на примере системы, состоящей из двух микропроцессорных систем (МПС), включающих в себя четыре однородных процессора. Поток заданий на обработку является пуассоновским со значением среднего интервала 5 ед.времени. МПС осуществляют многоканальное обслуживание заданий. Время обработки задания процессором МПС равно 3 ед.времени. В среднем 70% заданий обрабатывается первой МПС, остальные - второй. GPSS- модель параллельной системы представлена на рис.6.2.

```

MPC1     STORAGE 4
MPC2     STORAGE 4
EXPON FUNCTION RN1,C24
0,0/.1,.104/.2,.222/.3,.355/.4,.509/.5,.69/.6,.915
.7,1.2/.75,1.38/.8,1.6/.84,1.83/.88,2.12/.9,2.3
.92,2.52/.94,2.81/.95,2.99/.96,3.2/.97,3.5/.98,3.9
.99,4.6/.995,5.3/.998,6.2/.999,7/.9997,8
GENERATE 5,FN$EXPON
TRANSFER .7,K2,K1
K1 QUEUE  BUF1
ENTER    MPC1
ADVANCE  3
LEAVE   MPC1
DEPART  BUF1
TRANSFER ,KOH
K2 QUEUE  BUF2
ENTER    MPC2
ADVANCE  3
LEAVE   MPC2
DEPART  BUF2
KOH TERMINATE 1
START   1000

```

Рис.6.2. Моделирование параллельной СМО.

### ЗАДАНИЕ

1. Запустить программы 6.1 и 6.2. Получить набор выходных статистических данных.

2. Изменить программу 6.1 для моделирования последовательной системы тремя процессорами и различными законами распределения времени обработки задач.

3. Изменить программу 6.2 для моделирования параллельной системы с тремя МПС и получить набор выходных статистических систем.

4. С помощью статистических данных, полученных в результате моделирования построить зависимости вероятностно-временных характеристик последовательной и параллельной систем при различных законах распределения времени обработки задач и МПС.

## **СОДЕРЖАНИЕ ОТЧЕТА**

Отчет должен содержать:

1. Программы имитационного моделирования последовательных и параллельных систем при различных законах распределения времени обработки задач и МПС.

2. Графики зависимости и результаты анализа вероятностно-временных характеристик последовательных и параллельных систем.

## **КОНТРОЛЬНЫЕ ВОПРОСЫ**

1. Принципы работы последовательной СМО.
2. Принципы работы параллельной СМО.
3. Алгоритм последовательного обслуживания.
4. Алгоритм параллельного обслуживания.
5. Сравнительный анализ характеристик последовательной и параллельной СМО.
6. Примеры использования последовательных и параллельных систем передачи информации.

## ПРАКТИЧЕСКАЯ РАБОТА №7.

### МОДЕЛИРОВАНИЕ СЕТЕЙ МАССОВОГО ОБСЛУЖИВАНИЯ

**Цель работы.** Моделирование и анализ характеристик сетей массового обслуживания.

1. Исходные данные. Имеется сеть массового обслуживания (рис. 7.1).

Интервалы времени между поступлениями на вход системы заявок и длительности обслуживания заявок в системе распределены по экспоненциальному закону. Среднее значение интервала времени между поступлениями заявок равно 30.




Рис7.1. Исследуемая сеть массового обслуживания..

Первая фаза обслуживания заявки в исследуемой системе характеризуется следующим. Размер входного буфера `intBuff` равен 10. Длительность обслуживания заявки обслуживающим прибором `intProc` постоянна и равна 5.

Вторая фаза обслуживания заявки в системе характеризуется следующим. Размеры буферов `Buff1` и `Buff2` одинаковы. Обслуживающие приборы `Proc1` и `Proc2` имеют одинаковые характеристики.

После прохождения заявкой первой фазы обслуживания система определяет какой из буферов (Buff1 или Buff2) менее заполнен и направляет заявку в него для последующей обработки соответствующим обслуживающим прибором. Если буфер Buff1 заполнен полностью, все заявки будут направляться в буфер Buff2 вне зависимости от его состояния.

В случае если буфер intBuff, Buff1 или Buff2 заполнен полностью, вновь поступающие заявки отбрасываются.

2. Исследование системы. Для исследования указанных систем можно воспользоваться моделью на языке GPSS, приведенной в рис.7.2.

```

intBuffSize          VARIABLE  10
intSrvTime          VARIABLE  5
extBuffSize         VARIABLE  5
extSrvTime          VARIABLE  20
intTime               TABLE     M1,0,1,50
p1Time                TABLE     M1,0,2,90
p2Time                TABLE     M1,0,2,90

input  GENERATE(Exponential(1,0,30))
       TEST L   Q$intBuff,V$intBuffSize,out
       QUEUE   intBuff
       SEIZE   intProc
       DEPART  intBuff
       ADVANCE V$intSrvTime
       RELEASE intProc
       TABULATEintTime

part1  TEST L   Q$Buff1,Q$Buff2,part2
       TEST L   Q$Buff1,V$extBuffSize,out
       QUEUE   Buff1
       SEIZE   Proc1
       DEPART  Buff1
       ADVANCE(Exponential(1,0,V$extSrvTime))
       RELEASE Proc1
       TABULATEp1Time
       TRANSFER ,out

part2  TEST L   Q$Buff2,V$extBuffSize,out
       QUEUE   Buff2
       SEIZE   Proc2
       DEPART  Buff2
       ADVANCE(Exponential(1,0,V$extSrvTime))
       RELEASE Proc2

```

```

out      TABULATE p2Time
        TERMINATE 1
        START      1000000

```

Рис.7.2. Моделирование сети массового обслуживания.

Параметры системы имеют следующие значения:

- **intBuffSize** – размер буфера intBuff;
- **intSrvTime** – время обработки заявки обслуживающим прибором intProc;
- **extBuffSize** – размер буферов Buff1 и Buff2;
- **extSrvTime** – среднее время обслуживания заявки обслуживающими приборами Proc1 и Proc2.

3. Результаты исследования модели записать в таблицу 7.1.

Таблица 7.1.

Средние длины очередей при различных значениях среднего времени обслуживания заявки

Среднее время обслуживания	Средняя длина очереди intBuff	Средняя длина очереди Buff1	Средняя длина очереди Buff2
10			
15			
20			
25			
30			

Коэффициенты использования обслуживающих приборов при различных значениях среднего времени обслуживания заявки

Среднее время обслуживания	Коэффициент использования intProc	Коэффициент использования Proc1	Коэффициент использования Proc2
10			
15			
20			
25			
30			

## СОДЕРЖАНИЕ ОТЧЕТА

Отчет должен содержать:

1. Программы имитационного моделирования сетей массового обслуживания в соответствии с вариантом.
2. Графики зависимости и результаты анализа вероятностно-временных характеристик сетей массового обслуживания.

## КОНТРОЛЬНЫЕ ВОПРОСЫ

1. Основы теории сетей массового обслуживания.
2. Классификация сетей массового обслуживания.
3. Замкнутые сети массового обслуживания.
4. Разомкнутые сети массового обслуживания.
5. Методы исследования сетей массового обслуживания.
6. Особенности моделирования NGN на основе теории сетей массового обслуживания.

## СПИСОК ЛИТЕРАТУРЫ

1. Шеннон Р. Имитационное моделирование систем - искусство и наука: Пер. с англ. - М.: Мир, 1978.
2. Максимей И.В. Имитационное моделирование на ЭВМ. - М.: Радио и связь, 1988.
- 3.. Шрайбер Т.Дж. Моделирование на GPSS: Пер. с англ. - М.: Машиностроение, 1980.
4. Боев В.Д. Моделирование систем. Инструментальные средства GPSS WORLD: Учебное пособие.- СПб.: БХВ-Петербург, 2004.
5. Основы моделирования на GPSS/PC (часть 1). Методические указания для слушателей ФПКП / Л.А.Воробейчиков, Г.К.Сосновиков/МИС.-М.:1994.
6. Имитационное моделирование на языке GPSS. Методические указания /О.Н.Евсеева, В.В.Шишкин. - Ульяновск: УлГТУ, 1995.
7. Сети связи. Методические указания/Н.В.Гольцова.- Санкт-Петербург: СБГУТ, 2002.

## ОГЛАВЛЕНИЕ

ВВЕДЕНИЕ .....	3
1. ПРАКТИЧЕСКАЯ РАБОТА №1. СИСТЕМА ИММИТАЦИОННОГО МОДЕЛИРОВАНИЯ GPSS .....	4
2. ПРАКТИЧЕСКАЯ РАБОТА №2. МОДЕЛИРОВАНИЕ ОДНОКАНАЛЬНЫХ СМО .....	20
3. ПРАКТИЧЕСКАЯ РАБОТА №3. МОДЕЛИРОВАНИЕ МНОГОКАНАЛЬНЫХ СМО .....	32
4. ПРАКТИЧЕСКАЯ РАБОТА №4. МОДЕЛИРОВАНИЕ ПРИОРИТЕТНЫХ СМО .....	40
5. ПРАКТИЧЕСКАЯ РАБОТА №5. МОДЕЛИРОВАНИЕ СМО С ИЗМЕНЯЕМОЙ ИНТЕНСИВНОСТЬЮ ОБСЛУЖИВАНИЯ ...	44
6. ПРАКТИЧЕСКАЯ РАБОТА №6. МОДЕЛИРОВАНИЕ МНОГОФАЗНЫХ СМО .....	48
7. ПРАКТИЧЕСКАЯ РАБОТА №7. МОДЕЛИРОВАНИЕ СЕТЕЙ МАССОВОГО ОБСЛУЖИВАНИЯ .....	52
СПИСОК ЛИТЕРАТУРЫ .....	56
ОГЛАВЛЕНИЕ .....	57

Методические указания  
к практическим работам по предмету  
«Теория массового обслуживания в телекоммуникациях»  
для магистрантов по специальности 5А311301  
«Устройства и системы передачи информации»

Рассмотрены и рекомендованы  
к изданию на заседании  
научно-методического совета ТУИТ  
от «\_\_\_» \_\_\_\_\_ 2012 г.  
Протокол № \_\_\_\_\_

Составитель

Амирсаидов У.Б.

Ответственный редактор

Р.Х. Джураев

Корректор

Хамдам – Зода Л.Х.